

# SAS Technical News

Winter 2003

*For Higher  
Customer Satisfaction,  
We Bridge  
the SAS System  
Between  
Customer's World.*

## CONTENTS

- 1 特集 Scalable Performance Data Server  
~ユーザーの快適な環境を支えるデータサーバー~

---

- 7 Q&A

---

- 12 SASトレーニングのお知らせ

---

- 13 新刊マニュアルのご紹介

---

- 14 新教材販売開始のお知らせ

---

- 15 最新リリース情報

## 特集

# Scalable Performance Data Server

~ユーザーの快適な環境を支えるデータサーバー~

### はじめに

今回ご紹介するSPD Server( Scalable Performance Data Server ) は、マルチユーザー対応のデータサーバーで、数多くあるSASシステムのプロダクトの中でも、特に、DWH( データウェアハウス )内に存在するデータの処理をつかさどるものとして位置づけられています。対象としては、比較的規模の大きなデータを扱っているサイト、特に、処理全体の中でディスクのI/O( 入出力 )にそのほとんどの時間を取られているサイトが挙げられます。このようなサイトにおいて、処理のボトルネックがディスクのI/Oにあると思われる環境、あるいはハードウェアの性能が足りながらもそれを十分に活用できていない環境などでは、このSPD Serverを導入することによって、大幅なパフォーマンスの向上を図ることが可能となります。

また、個々のユーザーにおいても、日々扱うデータの量は増大し、その一方でマシンの性能も向上しています。こうした状況を受けて、SASシステムバージョン9では、データへのアクセスに関する部分、すなわちエンジンにおいて、SPD Serverの持つ多彩な機能の一部を搭載することになりました。バージョン9のSPDEエンジン( Scalable Performance Data Engine ) は、SPD Serverと同様に、ユーザーが快適にデータを処理することに主眼を置いています。今回の特集では、SPDEエンジンの詳細よりも、SPD Serverの基本的な動作や、SPD ServerとSPDEエンジンの違いを中心に解説します。これらについて理解しておくことは、今後SASシステムバージョン9を使用して多様なデータを処理する際に、大きな助けとなるに違いありません。

なお、2002年12月現在、SPD Serverの最新バージョンは3.0( TSM3 )で、SASシステム リリース8.2に対応しています。また、サポートされている環境は、次のとおりです。

ハードウェア	プラットフォーム
SMP対応	UNIX( AIX, HP-UX, Solaris, Compaq Tru64 )
	Windows NT, 2000

今回は、このSPD Serverについて、その機能を中心に次の順に説明します。後半では、SPD Serverと関わる範囲で、バージョン9のSPDEエンジンについても簡単に触れることになります。

- ・ SPD Serverの概要
- ・ データサーバーとしての機能
- ・ クライアントとサーバーの構成
- ・ SPD ServerとSPDEエンジン

## 2. SPD Serverの概要

### 2.1. 導入に適した環境

SPD Serverは、データの規模としては比較的大きなものを対象としていますが、機能の面ではデータの分析処理やクエリに対して最適化されています。そのような用途に特化して機能を絞り込むことにより、SPD Serverのパフォーマンスはデータを参照するときに最大限に引き出されます。たとえば、データの更新は週次/月次などの定期的なバッチ処理に限られる、というデータを持つ環境などには適しています。このことは、逆に、頻繁な作成あるいは更新を必要とするデータには適さない、ということの意味しています。SPD Serverは、いったんデータを格納した後に、その真価を発揮します。

### 2.2. ハードウェアの設定

SPD Serverの機能を活かすためには、SMP(Symmetric Multiprocessing)対応のハードウェアおよびオペレーティングシステムが必要です。SMPとは、1台のマシンに複数のCPUが搭載されたシステム形態を指します。この環境を前提として、SPD Serverは処理を分散し、パフォーマンスを向上させることができます。また、後述するように、SPD Serverは、クライアント/サーバーモデルを構成して持つので、クライアント側とサーバー側とのどちらにもソフトウェアをインストールする必要があり、サーバー側では、インストール後に若干の設定も必要となります。さらに、SASの他のプロダクトと異なり、ハードウェアやオペレーティングシステムに対しても、適切な設定が必要です。SPD Serverは、ディスクのI/O、すなわちハードウェアに大きく関わるため、ハードウェアの設定次第で、パフォーマンスの伸びは大きく変わります。したがって、導入の際には、SPD Serverによるファイルの分散をふまえたシステムによるRAID(ストライピングやミラーリングなど)の構成、またハードウェアに合わせたオペレーティングシステムの設定変更(UNIXにおけるカーネルの調整など)、といったシステム側に関する検討や配慮が必要となります。

### 2.3. データへのアクセス

SPD Server上のデータに対するアクセスは、基本的にBase SASソフトウェア上におけるデータセットに対してのそれと同様です。外観上は、Base SASソフトウェアのデータセットもSPD Serverのテーブルもほぼ同じです(ここでは、一般的なデータセットとSPD Serverのデータセットを区別するために、便宜的に「テーブル」という言葉を使用しています)。プログラム作成時も、Base SASソフトウェアと同じようにLIBNAMEステートメントでライブラリ参照名を割り当て、同じようにDATAステップやプロシジャを記述することができます。また、後述のとおり、SQLプロシジャはSPD Serverに対するパススルーと呼ばれる機能もサポートしています。この書式も、SAS/ACCESSソフトウェアと同様、基本的に業界標準の規格SQL92に準拠しています。並列処理やACL(アクセスコントロールリスト)など機能の一部は、SPD Server独自の仕様であるため特別な書式になりますが、主に利用するのは管理者に限られます。エンドユーザーにとっては、SPD Serverへのアクセスに対し、特別な意識は不要です。ここで、データへのアクセスが容易であることを示す例として、Base SASソフトウェアのデータセットをSPD Serverのテーブルに変換するプログラムを挙げます。次のように、LIBNAMEステートメントでそれぞれのエンジンを指定して、COPYプロシジャを使うことで、簡単に変換することができます。

Base SASソフトウェアで作成したデータセットをSPD Serverのテーブルに変換

```
/* V8エンジンで、SASデータセットのためのライブラリを割り当てます。*/
libname v8lib '/sasdata';

/* SASSPDSエンジンを指定して、SPD Serverのドメインに対して */
/* ライブラリを割り当てます。*/
libname spdlib sasspds 'tmp'
server=spdsrv.5150
user='spduser' password='xxxxxxx';

/* プロシジャの書式は、通常と同じです。*/
proc copy in=v8lib out=spdlib;
run;
```

なお、上のプログラムの中で使用している「SASSPDSエンジン」については、「4.2. クライアント用のエンジン」で解説します。また、「tmp」(グレイの部分)を指してSPD Serverの「ドメイン」という言葉を使用していますが、これについては「3.2. ファイルの分散」で改めて解説します。


## 3. データサーバーとしての機能

すでに触れているように、SPD Serverを使用する目的は、データを処理する際のパフォーマンスの向上にあります。この章では、パフォーマンス向上を実現する機能、またデータサーバーとして持つ周辺の機能について、掘り下げて説明します。

### 3.1. 並列処理


並列処理とは、複数のデータ処理を同時に並行して行なうことを意味し、SPD Serverの持つ機能のうちでも最も基本的で重要なものです。この機能は、複数のスレッド(マルチスレッド)を生成することで実現されます。スレッドとは、プロセスよりも小さなプログラムを処理するための機構です。SPD Serverは、スレッドで動作することによってハードウェアの多様な構成にも柔軟に対応できます。Base SASソフトウェアは、基本的に1つのプロセスにつき単一のスレッド(シングルスレッド)で動作するために、マシンが複数のCPUを搭載している場合でも、SASシステムを複数起動しない限りは、複数のマシンを有効に利用することができません。一方SPD Serverでは、複数のスレッドそれぞれが複数のCPU上で処理を分担して実行することによって、処理をより短時間で完了することができます。また、複数のCPUを同時に利用できるため、ディスクのI/Oもそれに応じて並行させられるため、後述のファイルの分散にも寄与し、パフォーマンスの向上をもたらします。次の図は、プロセスとスレッドの関係、またそれらとハードウェアの関係を示しています。

マルチスレッド



このように、複数のスレッドを生成することで、それらが複数のCPU間で分散され、またディスクに対して効率的にアクセスできます。ただし、ディスクに対するI/Oは、実際にはRAIDやディスクコントローラーの構成も関係するため、CPUごとにディスクが1対1に対応するということにはなりません。次に、実際にエンドユーザがSASシステムからSPD Serverに対してアクセスする上で、このマルチスレッドがどのような利点を持つか、ということについて説明します。たとえば、次の図では、Base SASソフトウェア上で作成したデータセットをSPD Server上にテーブルとして保存し(処理1)、次にそのテーブルからある条件のもとに抽出を行ない(処理2と処理3)、最後にその抽出結果を別のテーブルとして保存する(処理4)、という一連の処理を想定しています。ここで、抽出についてはテーブルの前半(処理2)と後半(処理3)というように作業を分担して同時に行ない、それらの結果を最後にまとめる(処理4)と効率を上げることができる、ということに注目しましょう。

#### 並列処理




上の図が示しているように、作業を分担して並列的に処理しているのが、複数のスレッド(スレッド1とスレッド2)です。先の「マルチスレッド」の図と照らし合わせると、これらがそれぞれ別のCPUに対応することも分かります。ユーザーからは、マルチスレッドによって並列処理が行なわれていることは見えませんが、処理時間は短縮されます。SASシステムバージョン8以降、SAS/CONNECTソフトウェアの新機能として、MP CONNECT(マルチプロセスCONNECT)が実装されています。この機能では、スレッドではなくプロセスを複数生成することによって、並列処理が実現されています。この場合は、1つまたはいくつかのDATAステップやプロシジャを単位とし、並列化を行ないます。一方、SPD Serverがサポートするマルチスレッドでは、1つのDATAステップやプロシジャの中で並列処理がなされます。この違いを把握した上で、これらの機能を併用することは処理時間の短縮のためには非常に有効です。

#### 3.2. ファイルの分散

マルチスレッドとともにパフォーマンスの向上に大きく貢献するのが、ファイルを分散する機能です。通常、Base SASソフトウェアでは、1つのデータセットが、ファイルシステム上の1つのファイルに対応します。これに対して、SPD Serverでは、1つのテーブルが設定に応じて自動的に複数のファイルに分割され保存されます。このようにファイルを複数に分けることによって、並列処理の際に、各ファイルに対して同時にアクセスし、処理を行なうことが可能となります。次の図は、ファイルシステム上に用意された複数の物理パスに対して、SPD Serverによって、どのようにファイルが分散されていくかを示します。1つのテーブルをディスクに書き込んだときに、そのデータの大きさと、あらかじめ設定したファイルあたりのサイズに従って、複数のファイルに分けて格納されます。次の図は、先の「マルチスレッド」の図について、下方のディスク部分を拡大したものにあたり、それぞれが「ディスク1」等を表しています。例えば、データのサイズが92MB(メガバイト)、ファイルあたりのサイズが16MBとした場合、このように書き表すことができます。

#### ファイルの分散



SPD Serverは、データの書き込みについては、並列処理をしません。上の図では、「データ1」のファイル、次に「データ2」のファイルというように、「データ6」のファイルまで順に作成していきます。ここでは、「データ1」から「データ5」が16MB、「データ6」が12MBで、合計92MBとなります。このように複数のディスクにまたがって作成することにより、読み込み時には、並列処理が可能となります。なお、大規模な構成の場合には、上の図で1つのディスクとされているものが、さらにRAIDによって複数のディスクから構成されることもあります。ところで、上の図では、データのためのディスクが4つ、メタデータとインデックスのためのディスクが1つずつとなっていますが、これはSPD Serverを起動する際に用意するファイルの中にあるドメインの定義に基づきます。たとえば、次のように、データのパスに4つの物理パスを記述し、そのそれぞれが別のディスクを指している場合には、上の図のような動作となります。

#### SPD Serverのドメイン定義

```
libname=tmp pathname= /spddata/meta
roptions="datapath=( '/spddata/data1' '/spddata/data2'
                    '/spddata/data3' '/spddata/data4' )
indexpath= '/spddata/index'";
```

なお、先の図にあった「メタデータ」とは、SPD Serverのテーブルについての管理情報です。また、「インデックス」については「3.4. インデックス」で詳述します。

#### 3.3. SQLパススルー

SAS/ACCESSソフトウェアでは、パススルーと呼ばれる機能によって、SQLで記述されたクエリをそのままデータベース(RDBMSなど)に引き渡し、データベース側で処理を行ない、結果のみをSASシステム側で受け取るということが可能です。これによって、クライアントとサーバーの間の無駄なデータのやり取りをなくすることができます。SPD Serverにおいてもこの機能が実装され、また処理によってはマルチスレッドによって処理されるので、より短時間で必要なデータを受け取ることができます。

## SQLパススルーのプログラム例

```
proc sql;
  connect to sasspds (
    dbq='tmp' server=spdsrv.5150
    user='spduser' password='xxxxxxx'
  );
  /* 括弧で囲われた部分が、サーバーに渡される部分です。 */
  select * from connection to sasspds (
    select * from tabl where str1='a' and num1>0
  );
quit;
```

## 3.4. インデックス

インデックスは、データを検索する際のパフォーマンスを向上させるために使われます。SPD Serverにおいても、Base SASソフトウェアと同様に、テーブル内の変数に対してインデックスを作成できます。インデックスの内部構造として、Base SASソフトウェアがB-treeという方式を採用しているのに対して、SPD ServerはB-treeとビットマップを組み合わせたハイブリッドインデックスという方式を採用し、マルチスレッドと合わせて、より高い検索効率を実現しています。また、インデックスには、作成時間がかかるという問題がありますが、SPD Serverは、インデックス作成の際もマルチスレッドでの並列処理を行なうため、より短時間で作業を完了することができます。

## インデックスを並列的に作成するプログラム

```
/* インデックスの並列的な作成を有効にします。 */
%let spdsiasy=YES;

proc datasets lib=spdlib;
  modify tabl;
  /* 単一インデックスVAR1とVAR2は、同時に作成されます。 */
  index create str1 num1;

  /* 複合インデックスIDX1とIDX2は、同時に作成されます。 */
  index create idx1=(str1 num1) idx2=(str2num2);
quit;
```

なお、SPD Serverは、インデックスを管理するためのユーティリティixutilを用意しています。このユーティリティにより、ディスク容量の節約と検索パフォーマンスの向上とを意図したインデックスの再構成、インデックス構造の確認などが容易に実行できます。

## 3.5. ソート処理

ソート処理においても、SPD Serverは指定に応じて並列処理を行いません。SPD Serverでは、ソートのアルゴリズムについて、ヒープソートとクイックソートのいずれかを指定する(SPDSTYPマクロ変数)ことができます。デフォルト(何も指定しない場合)は、ヒープソートとなります。それぞれのアルゴリズムに、長所と短所がありますが、並列処理をサポートしているのは、クイックソートのみとなっています。また、Implicitソートという機能を搭載することによって、ソートされていないテーブルに対してBYステートメントを指定した場合にも、暗黙的に(implicitly)ソートを実行できます。

## 3.6. ACL(アクセスコントロールリスト)

ACLとは、SPD Serverが独自に管理する、ユーザーによるデータに対するアクセス権を設定する機能です。これらは、オペレーティングシステムのユーザーや権限からは独立して扱われ、ライブラリ、テーブル、列(カラム)が権限設定の対象となります。管理者は、ユーザーまたはグループ単位で、読み込みや書き込みなどの許可を与えることができます。重要なデータが不用意に削除されることを未然に防いだり、特定のユーザーのみを対象としてデータを公開する、といったことを容易に実現することができます。次のプログラムでは、SPD Serverの管理用に用意されたSPDOプロシジャを使用して、テーブルに対してACLを設定しています。

## ACLを設定するプログラム

```
/* ACLの設定ではSPDOプロシジャを使用します。 */
proc spdo lib=spdlib;
  set acluser;

  /* テーブルを作成したユーザーと同じグループに
     属するユーザーにのみ参照を許可します。 */
  add acl tabl / groupread;
quit;
```

## 3.7. データサーバーの管理

その他、SPD Serverは、Base SASソフトウェアにはない、データ保守のための機能やユーティリティを用意しています。管理者はこれらを使用して、ユーザーアカウントの追加や削除、ファイルの整理、インデックスの保守などが実行できます。


## 4. クライアントとサーバーの構成

ここでは、SPD Serverが、サーバーとして具体的にどのような構成をとっているか、ということについて説明します。SPD Serverは、SAS/CONNECTソフトウェアやSAS/SHAREソフトウェアと同じように、クライアント/サーバー・モデルを採用しています。このモデルにより、コンピューター資源の分散が可能となるとともに、SPD Serverは、Base SASソフトウェアから独立して稼動することが可能になっています。

## 4.1. クライアント/サーバー・モデル

SPD Serverを、クライアント側とサーバー側というように機能的に分割することによって、それぞれの役割を活かすことが可能となり、クライアント側のエンドユーザーは、高性能のサーバーに処理を実行させることができます。これはまた、スタンドアロン環境で動作するSPDEエンジンと大きく異なる点でもあります。

## SPD Serverの構成



上記の図のように、クライアントマシンの側からSPD Serverへの接続形

態としては、次のものがあります。

- ・ 直接の接続
- ・ SAS/CONNECTソフトウェアを経由した、サーバー上のSASシステムからの接続
- ・ ODBCあるいはJDBCによる接続

また、その他の接続手段として、C言語に対応するAPIも提供されています。

#### 4.2. クライアント用のエンジン

クライアント側( Base SASソフトウェア側)には、SPD Serverに接続するためのエンジン、SASSPDSエンジンが用意されています。クライアント用のモジュールを設定した後は、V8エンジンなど、通常のBaseエンジンと同様にデータに対してアクセスすることができます。このエンジンを使用した接続方法は、次のとおりです。

ドメインに対するライブラリの割り当て(プログラム)

```
libname spdlib saspds 'tmp'
server=spdsrv.5150
user='spduser' password='xxxxxx';
```

上記のように、LIBNAMEステートメントのエンジンに「SASSPDS」(グレーの部分)を指定し、続けてオプションとして、ドメイン、サーバー、SPD Server用のユーザー等を指定します。次に、このときのログを見てみましょう。

ドメインに対するライブラリの割り当て(ログ)

```
NOTE: This is an production version of SPDS Engine
at SAS Version 8 level, TSM2
NOTE: User spduser(ACL Group SPDGRP) connected to
SPDS(HP-WIDE) 3.00 server at 192.168.1.1.
NOTE: ライブラリ参照名 SPDLIB を次のように割り当てました。
エンジン : SASSPDS
物理名 : :256/spddata/meta/
```

ログ上には、Base SASソフトウェアでの一般的なライブラリの割り当てとは多少異なった、一連の情報、ユーザーなど(グレーの部分)が表示されます。このとき、認証に失敗した場合には、その旨のエラーメッセージが表示されます。なお、上記のログの「/spddata/meta/」は、「3.2. ファイルの分散」でのドメイン定義の中にある、メタデータへの物理パスとなります。また、「256」は、このライブラリの処理を受け持つサーバープロセスのPID(プロセスID)となります。詳細については、「4.3.3. プロキシ(spdsbaseプロセス)」で説明します。

#### 4.3. サーバープロセス

サーバー側では、クライアントから処理を受けるためのサーバープロセスをあらかじめ起動しておきます。SPD Serverに關係するプロセスを表示してみましょう。

起動されているプロセス( HP-UXの出力を一部編集)

```
$ ps -ef | grep spds
spdadm 2501 1 16:30:00 spdsnsrv -listenport 5150 ...
spdadm 2503 1 16:30:05 spdserv -nameserverport 5150 ...
spdadm 2505 2503 16:30:05 spdslog 50985 ...
spdadm 2506 1 16:30:07 spdsnet -listenport 5151
spdadm 2507 2503 16:30:17 spdsbase ... 0.0.0.0 ...
```

この出力の中で、特に、次の3つのプロセス(それぞれグレーの部分)が、データサーバーとして機能するために重要な役割を果たしています。

- ・ spdsnsrvプロセス
- ・ spdservプロセス
- ・ spdsbaseプロセス

以下で、それぞれのプロセスについて、詳しく見ていくことにしましょう。

##### 4.3.1. ネームサーバー( spdsnsrvプロセス)

SPD Serverに接続するクライアントは、まずこのネームサーバーに接続します。つまり、このネームサーバーは、クライアントとデータサーバーを結ぶ窓口の役割を果たします。また、その一方で、データサーバーからは、そのサーバーが公開するドメインの登録を受け付けます。クライアントからの接続を受けた際には、ここで登録された情報をもとにして、ユーザーが指定するドメインを実際にデータが存在するファイルシステム上の物理パスに置き換えます。

##### 4.3.2. データサーバー( spdservプロセス)

このプロセスは、まず起動時にドメインの定義を読み取り、それをネームサーバーに登録します。その後は、SPD Serverのデータにアクセスするために待機していますが、実際には、プロキシ(代理)のプロセスを起動し、そのプロキシにユーザーとデータの橋渡しを行なわせます。また、プロキシを起動する前には、先に述べたACLを利用して、接続するユーザーを認証します。認証を受けたユーザーだけが、プロキシを経由してデータにアクセスできます。データサーバー自体は、認証とプロキシ起動を処理するだけなので、それ以外は常に待機状態にあり、複数のユーザーの接続を受け持つことができます。

##### 4.3.3. プロキシ( spdsbaseプロセス)

これは、データサーバーによって起動され、実際にデータにアクセスする役割を負ったプロセスです。ユーザーごとに独立してプロキシサーバーを立てることにより、あるプロキシに問題が発生した場合にも、他のユーザー、ドメインの接続には影響が出ません。このプロセスは、SPD Server起動時には見られませんが、ユーザーが接続した際に生成されます。接続後にSPD Serverに關するプロセスを確認してみましょう。

新たに起動されるspdsbaseプロセス( HP-UXの出力を一部編集)

```
$ ps -ef | grep spds
spdadm 2501 1 16:30:00 spdsnsrv -listenport 5150 ...
spdadm 2503 1 16:30:05 spdserv -nameserverport 5150
...
spdadm 2505 2503 16:30:05 spdslog 50985 ...
spdadm 2506 1 16:30:07 spdsnet -listenport 5151
spdadm 2507 2503 16:30:17 spdsbase ... 0.0.0.0 ...
spdadm 2601 2503 16:35:00 spdsbase ... 192.168.1.1 ...
```

この出力に見られるように、サーバー起動時に生成されるspdsbaseプロセスに加えて、もう1つspdsbaseプロセス(グレーの部分)が新たに起動されています。また、ここでは、IPアドレスが192.168.1.1(グレーの部分)のクライアントからの接続であることが分かります。このプロセス(プロセスIDは2601)が機能しなくなったとしても、spdserv(プロセスIDは2503)自体は影響されないため、他のクライアントからの接続を受け続けることができます。

4.3.4. SNETサーバー( spdssnetプロセス)

このサーバーは、ODBCやJDBCを経由してクライアントから接続する際に使われます。これによって、クライアント側にSASシステムを導入していない場合でも、SPD Serverを利用することができます。ただし、SPD Serverは、クライアントとしてSASシステムを利用するように最適化されているので、最大限のパフォーマンスを得るためには、クライアント側にSASシステムを導入する必要があります。この形態の接続を利用しない場合には、このサーバーを起動しておく必要はありません。

4.3.5. ログの管理( spdslogプロセス)

その他、これまでの諸サーバープロセスとは別に、SPD Serverに対するアクセスのログを記録するために、spdslogプロセスが起動されます。このプロセスによって、所定の場所にログを書き出されるので、管理者は、いつどのユーザーによって、どのようなアクセスがあったのかを確認することができます。たとえば、次のような一連のアクセスがあった際に、ログに書き出される情報を見てみましょう。

1. あるクライアントがライブラリを割り当ててデータサーバーへ接続
2. テーブルの作成
3. 作成したテーブルに対するACLの設定
4. そのテーブルの読み込み
5. サーバーからの切断

SPD Serverのログ

```


12/09/02 15:22:00:(596) SPDSBASE- AUDIT:
Lib=/spddata/ Res=*LIBNAM*.DATA
Owner=( _NONE_ . _NONE_ ) User=SPDUSER
Access=(ASSIGN.ACCESS) Perms=(READ.READ) ACL=N
12/09/02 15:22:06:(596) SPDSBASE- AUDIT:
Lib=/spddata/ Res=TAB1.DATA Owner=(SPDUSER.)
User=SPDUSER Access=(OPEN.OUTPUT)
Perms=(WRITE.WRITE) ACL=N
12/09/02 15:22:32:(588) SPDSSERV- OPER client
connection from IP address: 192.168.1.1
12/09/02 15:23:35:(588) SPDSSERV- SPDS OPER client
disconnected
12/09/02 15:23:35:(588) SPDSSERV- SPDS OPER client
IP address: 192.168.1.1
12/09/02 15:23:37:(596) SPDSBASE- AUDIT:
Lib=/spddata/ Res=TAB1.DATA Owner=(SPDUSER.)
User=SPDUSER Access=(OPEN.INPUT) Perms=(READ.READ)
ACL=Y
12/09/02 15:23:41:(596) SPDSBASE- SAS client with
id=SPDUSER has disconnected
12/09/02 15:23:41:(596) SPDSBASE- SAS client
connection is terminated
12/09/02 15:23:41:(588) SPDSSERV- SPDS User
spduser, OS User spdadm, at node=localhost has
disconnected
12/09/02 15:23:41:(588) SPDSSERV- SPDSBAS PID = 596
terminated
    
```

5. SPD ServerとSPDEエンジン

すでに述べたように、SASシステムバージョン9からBase SASソフトウェアにもSPD Serverの一部の機能を持ったSPDEエンジンが搭載されます。まずは、SPD ServerとこのSPDEエンジンとの違いを説明します。

まず、構成面における違いとして、SPD Serverがクライアント/サーバー・モデルをとっているのに対して、SPDEエンジンは、スタンドアロン環境に特化されている、ということ挙げることができます。したがって、SPDEエンジンでは、それに対応するサーバーを起動するということはありません。処理を実行させる側とその処理を実際に行う側という区別もありません。その分、個人で気軽に使用することができ、また複雑な管理も不要になる、という利点があります。位置づけとしてはBaseエンジンを拡張するものとなります。

SPDEエンジンを使用する構成



また、機能面から見た場合、SPDEエンジンは、SPD Serverの機能の一部を省略したものとなっています。主要な機能である、マルチスレッドや、ファイルの分散をサポートしているので、パフォーマンスの向上は十分期待することができます。ただし、SQLパススルーやACLといった機能は、クライアント/サーバー・モデルではないということもあり、省かれています。

SPD ServerとSPDEエンジンの比較一覧

	SPD Server	SPDEエンジン
構成	クライアント/サーバー	スタンドアロン
ファイルの分散		
SQLパススルー		×
パラレルWHERE		
パラレルグループBY		×
ハイブリッドインデックス		
Implicitソート		
ACL		×
ロックレベル	レコード	メンバー
カタログ		×
バックアップ		×

6. おわりに

今回ご紹介したSPD Serverは、SASシステムの中で、分析処理や参照に使われるデータをユーザーに円滑に提供するリポジトリとして機能します。このソフトウェアを導入して有効に活用することによって、目的の結果を快適に引き出すことができます。ここで快適というのは、エンドユーザーに不必要な待ち時間を与えないことであり、その点から、SPD Serverは、SASが提供する多くのプロダクトを裏で支える役割を担います。標準のBaseエンジン、SPD Serverに加え、バージョン9のSPDEエンジンが登場し、SASシステムが保持するデータの形態は多様化します。今後、用途に応じてこれら3種の中から最適なものを選択することの重要性は高まります。これら使い分けることで、業務に必要とされる作業の時間を短縮し、エンドユーザーに無駄な障害を感じさせないことが、システムの管理者に対しては求められることとなるでしょう。SPD ServerおよびSPDEエンジンに関する情報は、以下の弊社および米国SASインスティテュート社のホームページから参照できます。

- SPD Serverの概要(日本語)  
[http://www.sas.com/japan/software/sp\\_data.html](http://www.sas.com/japan/software/sp_data.html)
- SPD Serverの概要(英語)  
<http://www.sas.com/products/spds/>
- SPDEエンジンの概要(英語)  
<http://www.sas.com/service/library/onlinedoc/spde/>

# Q&A

ストアードマクロの使用法

ストアードプログラムの作成方法

V8で外部ファイルを作成する場合、

値をダブルクォーテーションで囲んで出力する方法

%IFマクロプログラミングステートメントにて

小数点を含む式を評価する方法

LAG関数を使用したプログラムの注意点

値の順番に応じて一定の割合でオブザベーションを取り出す方法

GPLOTプロシジャで、複数の参照線の色や線種を

個別に指定する方法

Windows XP上でのSASシステムの利用について

SAS/AFソフトウェアで「名前を付けて保存する」の

ダイアログ画面を利用する方法

方程式の解法

ロジスティック回帰におけるタイプ3分析について

## Q

コンパイル済みのマクロを特定のディレクトリに保存しておき、後で呼び出して実行することはできますか。

## A

マクロをコンパイルしてライブラリに保存するためには、まずMSTOREDシステムオプションを指定して、コンパイル済みマクロ機能を有効にします。その上で、SASMSTORE=システムオプションを使用して、マクロを保存するライブラリ参照名を指定します。さらに、マクロの定義時にSTOREマクロステートメントオプションを指定することで、SASMSTORE = システムオプションで指定したライブラリ内にSASマクロカタログ「SASMCR」として保存することができます。

例1：コンパイル済みマクロ「sample」を「C:%temp」ディレクトリに保存する

```
LIBNAME mylib 'C:%temp';
OPTIONS MSTORED SASMSTORE=mylib;

%MACRO sample /STORE;
  PROC PRINT DATA=sashelp.class;
  RUN;
%MEND sample;
```

例2：例1で保存したコンパイル済みマクロ「sample」を呼び出す

```
LIBNAME mylib 'C:%temp';
OPTIONS MSTORED SASMSTORE=mylib;
%sample
```

## Q

ストアードマクロという言葉を知っていますが、DATAステップにおいても、あらかじめプログラムをコンパイルして保存しておくことは可能でしょうか。

## A

通常、DATAステップでプログラムをサブミットすると、SASシステムはプログラムをコンパイルし、その後に行いますが、ストアードプログラムはすでにコンパイル済みのため、コンパイルせずに直接実行することが可能です。したがって、多くのステートメントを含むDATAステップをストアードプログラムとして作成すれば、コンパイル時間を節約できるため、プログラムの実行時間が短縮されます。また、ストアードプログラムの実行時に、元のプログラムで指定した入・出力データセット名を変更することも可能です。定型的な処理を行なうDATAステップを頻繁に使用する場合には向いている機能といえます。以下は、ストアードプログラムの作成、および実行例です。

```
/** ストアドプログラムの作成 **/
      /* PGM=に作成するストアードプログラム名を指定 */
DATA store1 / PGM=makedata;
  SET sashelp.class;
RUN;

/** ストアドプログラムの実行 **/
      /* PGM=実行するストアードプログラム名を指定 */
DATA PGM=makedata;
RUN;

/** 出力データセット名を変更 **/
DATA PGM=makedata;
      /* store1(旧出力データセット) = store2(新出力データセット) */
  REDIRECT OUTPUT store1=sotre2;
RUN;
```

なお、入力データセット名を変更する場合は、REDIRECTステートメントにて以下を指定します。

```
REDIRECT INPUT 旧入力データセット = 新入力データセット;
```

その際、新しく指定する入力データセットに含まれる変数の数や属性は、元の入力データセットに含まれているものと一致している必要があります。

## Q

SASシステムバージョン8を使用して、SASデータセットの内容をCSVファイルにしています。その際、文字変数のデータに「(ダブルクォーテーション)」をつけた形で出力したいのですが、何かよい方法はありますか。

## A

SASシステムバージョン8から、FILEステートメントにDSDオプションを指定できるようになったため、CSVファイルを簡単に作成できます。なお、文字変数のデータに「(ダブルクォーテーション)」を追加するには、\$QUOTEフォーマットが便利です。次のサンプルプログラムを参考にしてください。

```
DATA test;
  INPUT key $ a1 a2;
CARDS;
001 1 1
;
RUN;

DATA _null_;
  SET test;
  FILE "c:\temp\test.csv" DSD;
  FORMAT key $QUOTE.;
  PUT key a1 a2;
RUN;
```

**Q** %IFマクロプログラミングステートメントにて、下記のように「2 > 12.0」を比較しています。

```
(中略)
%IF 2 > 12.0 %THEN %DO;
(中略)
```

結果には、当然FALSEを期待しているのですが、TRUEになってしまいます。理由と回避方法を教えてください。

**A** %IFマクロプログラミングステートメントでは、マクロ式の自動評価が行なわれるため(暗黙の%EVALマクロ関数を適用)、小数点を含む値が指定されていると、%IFマクロプログラミングステートメントが、マクロ式を文字列と判断します。そのため、FALSEを期待しているにも関わらず、結果はTRUEになってしまいます。小数点を含むマクロ式の評価を行なう場合には、以下のサンプルプログラム(マクロcheck)のように、%SYSEVALFマクロ関数を明示的に指定する必要があります。

例：%sysevalfマクロ関数の指定

```
OPTIONS MLOGIC;
%MACRO check(fnum,ope,snum);
  /* %sysevalfマクロ関数を使用 */
  %IF %SYSEVALF(&fnum &ope &snum) %THEN
    %DO;
      PROC PRINT DATA=sashelp.class;
      RUN;
    %END;
  %ELSE %DO;
      PROC CONTENTS DATA=sashelp.class;
      RUN;
    %END;
%MEND check;
%check(2,>,12.0)
```

**Q** LAG関数をIFステートメントの中で使用した場合、結果が期待と異なる場合があります。なぜでしょうか。

**A** この現象は、LAG関数は、キューの操作、つまり値の保存を「実行されたときだけ行なう」という性質を持っているためです。LAG関数は、nオブザベーション前の値を返すのではなく、LAG関数を実行するたびに蓄積されたキューのn個前の値を返します。nオブザベーション前の値を取得するには、LAG関数を毎回実行して全ての値をキューに蓄積する必要があります。プログラムを例に説明します。下記のプログラムは、「前月との気温差をdif、偶数月のときにはその直前の偶数月(その月の2月前)との気温差をdif2」と出力することを目的として作ったプログラムです。

例1

```
DATA lag ;
  INPUT chiku $ month kion ;
  dif = kion - LAG1(kion) ;
  IF MOD(month,2) = 0 THEN
    dif2 = kion - LAG2(kion) ;
DATALINES;
北海道 1 -20
北海道 2 -18
北海道 3 0
北海道 4 10
北海道 5 20
北海道 6 15
北海道 7 20
北海道 8 28
北海道 9 22
北海道 10 17
北海道 11 10
北海道 12 2
;
RUN;
```

例1で作成されたデータセットlagの内容

month	kion	dif	dif2
1	-20	.	.
2	-18	2	.
3	0	18	.
4	10	10	.
5	20	10	.
6	15	-5	33
7	20	5	.
8	28	8	18
9	22	-6	.
10	17	-5	2
11	10	-7	.
12	2	-8	-26

(期待していたdif2の結果)

.
-18 - (欠損値) = (欠損値)
.
10 - (-18) = 28
.
15 - 10 = 5
.
28 - 15 = 13
.
17 - 28 = -11
.
2 - 17 = -15

上記の結果からわかるように、dif2の結果は2ヶ月のものではなく、「2回前にLAG関数により保存された値との計算結果」となっています。



month	kion	dif2	(実際のdif2の計算)
1	-20	.	.
2	-18	.	-18 - .
3	0	.	.
4	10	.	10 - . (2回前のLAGは存在しない)
5	20	.	.
6	15	33	15 - (-18) (2回前のmonth=2で保存された値)
7	20	.	.
8	28	18	28 - 10 (2回前のmonth=4で保存された値)
9	22	.	.
10	17	2	17 - 15 (2回前のmonth=6で保存された値)
11	10	.	.
12	2	-26	2 - 28 (2回前のmonth=8で保存された値)

期待していると通りの結果を出力させる場合は、次のようなプログラムを実行します。

例2

```
DATA lag ;
  INPUT chiku $ month kion ;
  dif = kion - LAG1(kion) ;
  dif2 = kion - LAG2(kion) ;
  IF MOD(month,2) THEN dif2 = . ; /* 奇数月の場合は欠損値 */

DATALINES;
北海道 1 -20
北海道 2 -18
北海道 3 0
北海道 4 10
北海道 5 20
北海道 6 15
北海道 7 20
北海道 8 28
北海道 9 22
北海道 10 17
北海道 11 10
北海道 12 2
;
RUN;
```

例2で作成されたデータセットlagの内容

month	kion	dif	dif2
1	-20	.	.
2	-18	2	.
3	0	18	.
4	10	10	28
5	20	10	.
6	15	-5	5
.....(以下省略)			

**Q** あるデータセットから、値のもっとも小さいものから10%程度のオブザベーションを取り出したいのですが、どのような方法がありますか？

**A** SORTプロシジャを利用して取り出すことも可能ですが、ここではRANKプロシジャとPERCENTオプションを利用する方法をご紹介します。RANKプロシジャは、データセットの変数の値に順番をつけるプロシジャですが、PERCENTオプションを使うことで、百分率での順番をつけることができます。以下のプログラムでは、RANKプロシジャにより作成された順番を、抽出条件として利用しています。

```
/* サンプルデータ */
DATA sample;
  DO student = 1 TO 100;
    tensu = INT(RANUNI(0)*100+1);
  OUTPUT;
  END;
RUN;

/* 値の小さいものから 10% を抽出 */
PROC RANK DATA=sample OUT=outdata(WHERE=(rank<=10))
  PERCENT ;
  VAR tensu;
  RANKS rank;
RUN;
```



**Q** GPLOTプロシジャでVREFオプションを使って複数の参照線を描いています。参照線の色や線の種類を1本ずつ個別に指定できますか？

**A** VREFオプション、およびHREFオプションで描かれる参照線の色や線種を1本ずつ個別に指定することはできません。指定できる参照線の色と線種は、1種類です。しかし、ANNOTATE機能を使用すれば、1本ずつ参照線を描くことができます。下記の例を参考にしてください。

例：縦軸の100, 150, 200に参照線を描く

```

/* サンプルデータの作成 */
DATA a;
  INPUT x y @@;
CARDS;
10 77 20 200 30 230
15 150 25 210 35 210
;
RUN;

/* ANNOTATEデータセットの作成 */
DATA _anno;
  LENGTH function $8;
  RETAIN xsys '1' ysys '2' when 'a';
  FUNCTION='move'; x=0; y=100; OUTPUT;
  FUNCTION='draw'; x=100; y=100; line=10;
  color='blue'; OUTPUT;
  FUNCTION='move'; x=0; y=150; OUTPUT;
  FUNCTION='draw'; x=100; y=150; line=3;
  color='green'; OUTPUT;
  FUNCTION='move'; x=0; y=200; OUTPUT;
  FUNCTION='draw'; x=100; y=200; line=1;
  color='red'; OUTPUT;
RUN;

/* ANNO=ANNOTATEデータセット名 */
PROC GPLOT DATA=a ANNO=_anno;
  PLOT y*x;
RUN;
QUIT;
    
```

**Q** Windows XP上でSASシステムを使用できますか。

**A** Windows XPについては、現在出荷中のSASシステムリリース8.2 TS2M0にて正式サポートしています。Windows XPは、当初は正式サポート対象ではありませんでしたが、その後の稼働確認作業を経て、サポート対象オペレーティングシステムに変更されました。ただし、32ビットのデスクトップ版のみとなります(サーバー版はサポート対象としていません)。また、オペレーティングシステムの仕様により、Home Editionでは一部動作に制限が発生する場合がありますため、Professional Editionの利用を推奨します。

**Q** SAS/AFソフトウェアの画面から、利用者が作成したファイルなどを任意のフォルダ/ファイル名で保存できるようにしたいのですが、名前を付けて保存するようなダイアログ画面を利用するにはどうしたらよいでしょうか。

**A** このようなダイアログを表示するには、SCLステートメントでのFILEDIALOG関数を使用します。ファイルを保存するための書き出し用のプッシュボタンなどを作成して、その定義内で次のように指定します。

```
rc=FILEDIALOG('SAVEAS',入力値用変数,' ',' ','*.*');
```

以上の処理で、入力されたフォルダ名・ファイル名の戻り値が返る入力値変数名を参照して、ファイルを作成することが可能です。

**注意:**  
当該ダイアログ(この場合はSAVEASなので「名前を付けて保存」)は、ユーザーがフォルダとファイル名を選択可能とし、選択された内容を変数に返しますが、そのまま保存はしません。実際の保存操作は、別途戻り値を受け取った後に、保存するプログラムを実行する必要があります。

FILEDIALOG関数の詳細については、SASシステムのヘルプから、キーワード「FILEDIALOG」を入力して確認できます。

**Q** SASで方程式を解く方法を教えてください。

**A** 次の2つの方法で直接対応することができます。

1. SAS/IMLソフトウェアのPOLYROOT関数  
POLYROOT関数を用いると、代数方程式の複素数解を求めることができます。具体的には、次のようなプログラムになります。

```

**** サンプルプログラム ****/
proc iml;

/* x^3- 2*x^2 + 5*x + 9=0「^」の後ろの数字は累乗を表す*/
coeff={1 -2 5 9}; /* 次数の高い方から係数を記述。最後は定数項 */

sol=polyroot(coeff);
print sol;
**** ここまで ****/

**** 出力 ****/
/*解は3つ得られた。1列目が実部、2列目が虚部*/

s

-1.080584      0
 1.540292  2.4405594
 1.540292 -2.440559
**** ここまで ****/
    
```

実数解が1つ、(共役な)虚数解が2つ得られたことがわかります。

## 2. SAS/ETSソフトウェアのMODELプロシジャ

MODELプロシジャを用いると、連立方程式(線形、非線形)の「実数解」を求めることができます。次のプログラムは、連立方程式

$$\sqrt{x}-y=0$$

$$3+2/x-y=0$$

を解くものです。

```

/**** サンプルプログラム *****/
data test;
  input a b @@;
cards;
3 2
;
run;

/*MODELプロシジャ*/
proc model data=test;

/*方程式は、以下のようにeq.### = の形で記述する */
eq.sqrt      = sqrt(x) - y;
eq.hyperbola = a + b / x - y;
/* SOLVEステートメントで解くべき変数を指定 */
/* また、アウトプットウィンドウへ解を出力するため、SOLVEPRINTオプションも指定する*/
solve x y / solveprint;
run;
quit;
/**** ここまで *****/

/**** 出力 *****/
      Solution Values

           x           y

10.21329      3.19582
/**** ここまで *****/

```

なお、MODELプロシジャを利用した際には、解が2つ(2組)以上存在する場合であっても、1つ(1組)の解しか得られません。MODELプロシジャの詳細については、SAS/ETSソフトウェアのドキュメントをご参照ください。

**Q** 説明変数に交互作用を含むロジスティック回帰分析を行っています。その際に、CLASSステートメントのオプションとしてPARAM=REFを指定した場合と、PARAM=GLMを指定した場合のタイプ3分析(Type III Analysis of Effects)の結果が異なる場合があります。どうしてでしょうか。

**A** LOGISTICプロシジャにおいて交互作用を含めたタイプ3分析を実行すると、検定対象の帰無仮説が分類変数のパラメータ化によって異なります。そのため、表示される結果も違います。PARAM=REFオプションを指定した場合は、「主効果に対するパラメータが0である」という帰無仮説に対して検定を行いません。

このとき、主効果を含む交互作用に対するパラメータは考慮されません。一方、PARAM=GLMオプションを用いると、「主効果の水準に違いがない」という帰無仮説を検定することになります。この場合は、交互作用の効果を含めた上での検定を行なっています。以下に例として次のようなモデルを考えます。ここではyを応答変数とし、AとBをそれぞれ2水準('1'または'2')からなる分類変数(主効果)とします。主効果Aに対するパラメータはラベルがA1となりますので、このラベルを使ってTESTステートメントを指定します。2つのTESTステートメントはPARAM=REFオプションを指定した際のタイプ3分析と同じ結果が得られます。

```

proc logistic data=temp descending;
  class A B / param=ref;
  freq count;
  model y=A|B / scale=none aggregate;
  A : test A1; /*PARAM=REF 効果Aに対する検定*/
  B : test B1; /*PARAM=REF 効果Bに対する検定*/
run;

```

これに対し、PARAM=GLMを指定した際には、交互作用に対するパラメータのラベルA1B1も用いて以下のようにTESTステートメントを指定します。これにより、タイプ3分析と同じ結果を出力することが可能となります。

```

proc logistic data=temp descending;
  class A B / param=glm;
  freq count;
  model y=A|B / scale=none aggregate;
  /*PARAM=GLM 効果Aに対する検定*/
  A : test 2*A1+A1B1;
  /*PARAM=GLM 効果Bに対する検定*/
  B : test 2*B1+A1B1;
run;

```



# SAS Training

## SASトレーニングのお知らせ

### 特別コース開催のお知らせ

#### 「医学研究における経時データ解析」コース

日程: 3月19日(木)~20日(金) (東京会場)  
 価格: 80,000円(サービスチケット捺印数4)

テキストは英語版を使用します

受講対象: SASシステムによる統計解析に従事している方  
 前提知識: 「分散分析」コースを受講済みか、同程度の知識のある方  
 学習内容: 医薬統計専門家により、以下の内容で開催します。

多くの医学研究においては、研究対象者ごとに時間を追って反応変数を繰り返し測定することがあり、このような研究は経時観察研究などと呼ばれます。その研究目的の1つは、反応変数と時間、治療効果あるいは他の説明変数との関係を定量的に評価することです。同一対象者内の反応変数には相関が生じることから、統計的に妥当な推論を行なうためには測定単位に起因する相関を考慮する必要があります。そのような反応変数間の相関を考慮する方法として、線型混合モデルおよび一般化推定方程式(GEE)による解析方法が近年注目を浴びています。このコースでは、線型混合モデルと一般化推定方程式の理論的背景について講義し、実際の医学データの解析を通してMIXEDプロシジャやGENMODプロシジャの使用法について学びます。

- ・線型混合モデルおよび一般化推定方程式(GEE)に関する基本概念の説明
- ・MIXEDプロシジャやGENMODプロシジャの使用法
- ・医学分野における実データの解析

「マーケティング&マーケティング・リサーチ・コース」(全4回)  
 この春、下記の内容でマーケティングの業務に携わる方を対象としたトレーニングを開催する予定です。詳細については、弊社ホームページ(<http://www.sas.com/japan/training>)をご参照いただくか、弊社トレーニング担当までお問い合わせください。

日程: 3月13日(木)~4月17日(木) 全4回(東京会場)  
 価格: 120,000円(サービスチケット捺印数6)

前提知識: 「SASによるデータ解析入門」コースを受講済みか、同程度の知識のある方

#### 第1回: 3月13日(木)「マーケティング概論」

担当講師: 学習院大学 経済学部 上田隆穂 教授

学習内容: この回では、マーケティングとは何か、そしてリサーチとはどうつながっていくのかに重点をおいた解説を行ないます。特に分析関連の領域(新製品開発等)を詳しく解説していきます。この概論では、因子分析など一連のマップを描く手法から、クラスター分析、コンジョイント分析などの概要説明も含まれます。解説だけでなく冗長になりやすいのでコンピュータは使いませんが、演習形式での議論も含めることを予定しています。

第2回: 4月2日(水)「データ収集法とリサーチ手法、その解説と実践」

担当講師: 専修大学 商学部 奥瀬喜之 専任講師

学習内容: データの種類(1次データと2次データ)、1次データの収集方法(面接法、留置法、郵送法、インターネットによる方法など)、2次データの収集方法(インターネットを使った検索など)、およびリサーチの手法(定性調査と定量調査など)について説明します。また実践的に使える手法を実際に幾つか選んでコンピュータを用いた実習を行なっていきます。

#### 第3回(日程調整中)「リサーチ・プロセス、マーケティング・モデルの作り方」

担当講師: 日本大学 経済学部 齊藤嘉一 専任講師

学習内容: 問題に対してどのようなプロセスで対応していくか、リサーチ設計のプロセスを解説していきます。簡単なマーケティング・モデルをつくるまでのプロセス、その推定の仕方、アウトプットの見方、モデル修正の仕方を解説します。つまり、データ収集前の部分(マーケティング課題、リサーチ・クエスト、仮説)の解説から始まって分析、考察までを学んでいきます。モデルづくりの実践もコンピュータを用いた実習形式で行ないます。

日程についてはお問い合わせください。

#### 第4回: 4月17日(木)「分析事例」

担当講師: 学習院大学 経済学部 上田隆穂 教授

学習内容: この回では、実際にどのような問題意識において、どのような分析が行なわれたかを分析事例を通じて学んでいきます。分析の目的を設定し、そのためにはどのような分析プロセスをとるべきかを事前にディスカッションし、その後実際に行なった分析の解説を行なっていきます。この回で用いた分析手法は、知覚マップ、選好回帰、PSM、コンジョイント分析、クラスター分析、MDS、ニューラルネットワークを使った分析等です。手法自体よりも設定した分析テーマに対する分析の道筋、考え方を学んでいきます。

### トレーニングニュース(受講料値下げ)

#### 「臨床データマネジメント」コースおよび「解析表作成」コース アンケート実施

「臨床データマネジメント」コースおよび「解析表作成」コースについて、今後コースの改訂を企画しています。そのため、受講されるみなさまに、従来のトレーニング内容に加え、コース終了時に詳細なアンケートのご協力をお願いいたします。このアンケートの実施にあたり、従来の60,000円から40,000円に値下げしてコースをご提供します。改訂に向けたアンケートを実施するトレーニング日程は次のとおりです。

#### 「臨床データマネジメント」コース

東京会場 2月13日~14日、5月12日~13日

#### 「解析表作成」コース

東京会場 2月24日~25日、5月26日~27日

その他、トレーニングに関する情報については、弊社ホームページ(<http://www.sas.com/japan/training>)をご参照いただくか、下記<トレーニング担当>までお問い合わせください。

#### トレーニング担当

T E L 03-3533-3835

F A X 03-3533-3781

E-mail JPNTraining@sas.com

# New Publications

## 新刊マニュアルのご紹介

### 発売中のマニュアル

#### SASシステム リリース8.1の新機能(日本語版)

注文番号:10083

価格:3,500円(税抜)

本書は、SASシステム リリース8.1で追加、変更された機能についての解説書(「What's New in SAS Software for Release 8.1」、注文番号:58017)の翻訳版で、SASシステム リリース8.2を使用する場合も、本書の情報が必要となります。SAS/ETSソフトウェア、SAS/IMLソフトウェア、SAS/ORソフトウェア、SAS/QCソフトウェア、およびSAS/STATソフトウェアに関しては、さらに詳細な情報を解説した「Changes and Enhancements, Release 8.1(英語版)」が入手可能です(同内容のPDF版が、SASシステム リリース8.2導入時に提供される「What's New in SAS Release 8.1 and Release 8.2 Online Documentation」に含まれています)。なお本書では、原書記載のCMS版、OpenVMS版、OS/2版SASシステムおよびSAS/GISソフトウェアの章を省略しています。なお、本書と同内容のPDF版は、SASシステム リリース8.2導入時に提供される「SASシステムV8e(リリース8.2)インストールドキュメントメディア(CD-ROM)」に含まれています。

#### SASシステム リリース8.2の新機能(日本語版)

注文番号:10084

価格:3,500円(税抜)

本書は、SASシステム リリース8.2で追加、変更された機能についての解説書(「What's New in SAS Software for Release 8.2」、注文番号:58025)の翻訳版で、SASシステム リリース8.2を使用する場合は本書の情報が必要になります。SAS/ASSISTソフトウェア、SAS/ETSソフトウェア、SAS/IMLソフトウェア、SAS/ORソフトウェア、SAS/QCソフトウェア、およびSAS/STATソフトウェアに関しては、さらに詳細な情報を記載した「Changes and Enhancements, Release 8.2(英語版)」が入手可能です(同内容のPDF版が、SASシステム リリース8.2導入時に提供される「What's New in SAS Release 8.1 and Release 8.2 Online Documentation」に含まれています)。なお本書では、原書記載のCMS版、OpenVMS版、OS/2版SASシステムおよびSAS ADXインターフェイス、SAS/GISソフトウェアの各章を省略しています。なお、本書と同内容のPDF版は、SASシステム リリース8.2導入時に提供される「SASシステムV8e(リリース8.2)インストールドキュメントメディア(CD-ROM)」に含まれています。

### 2月以降発売予定のマニュアル

#### SAS言語:リファレンス編 バージョン8(仮題、日本語版)

注文番号:17239

価格:未定

本書は、バージョン8以降のBase SASソフトウェアにおけるSAS言語のリファレンスマニュアル(「SAS Language Reference: Dictionary, Version 8」、注文番号 57239)の翻訳版です。本書の内容は、「SASランゲージ リファレンス Version 6, First Edition」の「リファレンス編」のバージョン8対応版となり、下記について詳細に解説しています。

- ・SASデータセットオプション
- ・SAS出力形式
- ・SAS関数とSAS CALLルーチン
- ・SAS入力形式
- ・SASステートメント
- ・SASシステムオプション
- ・SAS言語におけるワードとSAS名に関する規則
- ・SAS演算子
- ・DATAステップデバッグ

SAS言語の基本的な仕様や概念の詳細は、「SAS言語:解説編 バージョン8」を参照してください。なお、SASシステム リリース8.2導入時に提供される「SASシステムV8e(リリース8.2)インストールドキュメントメディア(CD-ROM)」には、リリース8.2を使用する上で必要な「SASシステム リリース8.1の新機能」、「SASシステム リリース8.2の新機能」、「SAS言語:解説編 バージョン8」、および同内容のPDF版マニュアルが含まれる予定です。

#### SAS言語:解説編 バージョン8(仮題、日本語版)

注文番号:17375

価格:未定

本書は、バージョン8以降のSAS言語の仕様と概念についてのマニュアル(「SAS Language Reference: Concepts, Version 8」、注文番号: 57375)の翻訳版です。本書の内容は、「SASランゲージ リファレンス Version 6, First Edition」の「解説編」のバージョン8対応版となり、下記をはじめとする各項目について詳細に解説しています。

- ・SASシステムの基本概要
- ・SASシステムの処理
- ・SAS言語におけるワードや規則
- ・SASデータセット/システムオプション
- ・SAS入力形式/出力形式
- ・SAS変数とSAS CALLルーチン
- ・SASステートメント
- ・SAS変数
- ・欠損値
- ・SAS式
- ・SAS日付値・SAS時間値・SAS日時値
- ・エラー処理とデバッグ
- ・SAS出力
- ・BYグループやWHERE式の処理
- ・システムパフォーマンスの最適化
- ・2バイト文字セット
- ・DATAステップの処理
- ・生データの読み込み
- ・SASデータセットの読み込み・結合・変更
- ・配列処理
- ・ストアドプログラムの作成と実行
- ・DICTIONARYテーブル
- ・DBMSデータへのアクセス
- ・バージョン8と旧バージョンとの互換性
- ・ファイル保護
- ・SAS入出力エンジン

SAS言語の個々の言語要素の詳細については、「SAS言語:リファレンス編 バージョン8」を参照してください。なお、SASシステム リリース8.2導入時に提供される「SASシステムV8e(リリース8.2)インストールドキュメントメディア(CD-ROM)」には、リリース8.2を使用する上で必要な「SASシステム リリース8.1の新機能」、「SASシステム リリース8.2の新機能」、「SAS言語:リファレンス編 バージョン8」、および本書と同内容のPDF版マニュアルが含まれる予定です。

### SASロゴ入りグッズ毎月のキャンペーンご案内

毎月SASロゴ入りグッズを2アイテムピックアップし、キャンペーン価格にてご提供しています。キャンペーン内容は毎月変更されますので、お見逃しのないようお願いいたします。他にも多数取り揃えておりますので、ご興味のある方はぜひご利用ください。お申し込み方法は、マニュアル注文と同様です。

### SASロゴグッズのページ

<http://www.sas.com/japan/manual/goods.html>

### マニュアル販売係

T E L 03-3533-3835

F A X 03-3533-3781

E-mail [JPNBooksale@sas.com](mailto:JPNBooksale@sas.com)

SASマニュアル注文用紙、および最新のPublication Catalog

(マニュアル案内パンフレット)は弊社ホームページ

(<http://www.sas.com/japan/manual/>)にて公開しておりますので、併せてご利用ください。

# SAS Learning Edition Release 1.0

## 新教材発売開始のお知らせ

株式会社SASインスティテュートジャパンでは、このたび「SAS Learning Edition リリース1.0」を発売します。SAS Learning Edition リリース1.0とは、おもにSASシステムを個人学習していただくことを目的に、必要な機能・製品を抽出してご提供する教材パッケージです。

### SAS Learning Edition リリース1.0とは

SAS Learning Edition リリース1.0は下記の仕様を持ち、学生の方や新入社員の方の教材として、またビジネスシーンでも役立つ幅広い機能をカバーしています。

インターフェース	Enterprise Guideソフトウェア
使用可能またはインストールされるソフトウェア	Base SASソフトウェア、SAS/STATソフトウェア、SAS/QCソフトウェア、SAS/ETSソフトウェア、SAS/GRAPHソフトウェア

動作環境:Windows 95/98/Me/NT4.0/2000/XP  
利用できる機能や期間などに一部制限があります。制限の詳細については、弊社Webサイト(<http://www.sas.com/japan/manual/index.html>)をご参照ください。

### Web Based Trainingと併用して効率的な学習

SAS Learning Edition リリース1.0には、初めてSASをお使いになる方に好評の新トレーニングマテリアル「Web Based Training」を6ヶ月間利用できるIDも含まれています。

#### Web Based Trainingとは?

インターネットを利用して、ブラウザ上でSASシステムの機能や操作方法を学べる教育マテリアルです。SASの基本操作、ファイルの入出力方法、簡単なデータによるプログラムの作成、データ集計表の作成について、音声と画面でわかりやすく解説しています。インターネットに接続できる環境であれば、家庭でも、オフィスでも、学校でも、場所を選ばずにSASシステムを学習できます。

SAS Learning Edition リリース1.0のご購入お申し込みの翌月1日より(月の20日以降にお申し込みの場合は翌々月1日より)6ヶ月間、Web Based Training (Basicコース)を利用できます。Web Based Trainingで学んだ知識をSAS Learning Edition リリース1.0で実践することにより、さらに深く理解することができます。

### 発売受付開始と出荷時期

受付開始日 2003年1月27日(月)

出荷開始日 2003年2月17日(月) 予定

### 価格

SAS Learning Edition リリース1.0の発売を記念して、キャンペーン特別価格にて販売します。この機会にぜひお求めください。

#### [ キャンペーン期間 ]

2003年1月27日(月) ~ 3月31日(月) お申し込み分

一般利用: 通常価格42,800円  
キャンペーン期間特別価格 34,800円(税抜)  
アカデミック利用:通常価格30,800円  
キャンペーン期間特別価格 24,800円(税抜)

### お申し込み方法

SAS Learning Edition リリース1.0のご購入の際は、専用の注文用紙に必要事項をご記入の上FAXにてお申し込みいただくか、Webフォームよりお申し込みください。

専用の注文用紙やフォーム、また購入についての詳細は、弊社ホームページ(<http://www.sas.com/japan/manual/index.html>)をご覧ください。

#### マニュアル販売係

T E L 03-3533-3835

F A X 03-3533-3781

E-mail [JPNBooksale@sas.com](mailto:JPNBooksale@sas.com)

# Latest Releases

## 最新リリース情報

### PCプラットフォーム

Windows版 SASシステム リリース8.2 TS2M0

### ミニコンピュータプラットフォーム

OpenVMS AXP版 SASシステム リリース6.12 TS020

OpenVMS VAX版 SASシステム リリース6.08 TS407

### UNIXプラットフォーム

Tru64版 SASシステム リリース8.2 TS2M0

ABI+版 SASシステム リリース6.11 TS040

SunOS/Solaris版 SASシステム リリース8.2 TS2M0

HP-UX版 SASシステム リリース8.2 TS2M0

AIX版 SASシステム リリース8.2 TS2M0

Linux版 SASシステム リリース8.2 TS2M0

### メインフレームプラットフォーム

IBM版(MVS) SASシステム リリース6.09E TS470

富士通版(F4、MSP) SASシステム リリース6.09E TS470

日立版(VOS3) SASシステム リリース6.09E TS470

CMS版 SASシステム リリース6.08 TS410

## SAS Technical News Winter 2003

発行  
株式会社SASインスティテュートジャパン

テクニカルニュースに関するお問い合わせ先

テクニカルサポートグループ

TEL: 03-3533-3877

FAX: 03-3533-3781

E-mail: technews@jpn.sas.com



株式会社SASインスティテュートジャパン

[www.sas.com/japan/](http://www.sas.com/japan/)

東京本社  
〒104-0054  
東京都中央区勝どき1-13-1  
イヌイビル・カチドキ 8F  
Tel 03 (3533) 6921  
Fax 03 (3533) 6927

大阪支店  
〒530-0004  
大阪市北区堂島浜1-4-16  
アクア堂島西館 12F  
Tel 06 (6345) 5700  
Fax 06 (6345) 5655